Rules for integrands of the form $u \operatorname{Log}[e (f (a + b x)^p (c + d x)^q)^r]^s$

1: $\int u \operatorname{Log}\left[e \left(f (a + b x)^p (c + d x)^q\right)^r\right]^s dx$ when $b c - a d == 0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b c - a d == 0$, then $a + b x == \frac{b}{d} (c + d x)$

Rule: If $b c - a d == 0 \land p \in \mathbb{Z}$, then

$$\int u \operatorname{Log}\left[e \left(f (a + b x)^p (c + d x)^q\right)^r\right]^s dx \to \int u \operatorname{Log}\left[e \left(\frac{b^p f}{d^p} (c + d x)^{p+q}\right)^r\right]^s dx$$

Program code:

```
Int[u_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  Int[u*Log[e*(b^p*f/d^p*(c+d*x)^(p+q))^r]^s,x] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && EqQ[b*c-a*d,0] && IntegerQ[p]
```

2. $\int \operatorname{Log}\left[e \left(f (a + b x)^p (c + d x)^q\right)^r\right]^s dx$ when $b c - a d \neq 0$

2: $\int \operatorname{Log}\left[e \left(f (a + b x)^p (c + d x)^q\right)^r\right]^s dx$ when $b c - a d \neq 0 \land p + q \neq 0 \land s \in \mathbb{Z}^+ \land s < 4$

Derivation: Integration by parts

Basis: $1 == \partial_x \frac{a+b x}{b}$

Basis: $\partial_x \operatorname{Log}[e (f (a + b x)^p (c + d x)^q)^r]^s ==$

$$\frac{b r s (p+q) \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1}}{a+b x} - \frac{q r s (b c-a d) \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1}}{(a+b x) (c+d x)}$$

Rule: If $b c - a d \neq 0 \land p + q \neq 0 \land s \in \mathbb{Z}^+ \land s < 4$, then

$$\int \operatorname{Log}\left[e \left(f (a + b x)^p (c + d x)^q\right)^r\right]^s dx \to$$

$$\frac{(a+b x) \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{b} - r s (p+q) \int \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1} dx + \frac{q r s (b c-a d)}{b} \int \frac{\operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1}}{c+d x} dx$$

–

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  (a+b*x)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/b -
  r*s*(p+q)*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1),x] +
  q*r*s*(b*c-a*d)/b*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && NeQ[b*c-a*d,0] && NeQ[p+q,0] && IGtQ[s,0] && LtQ[s,4]
```

3. $\int (g+h\,x)^m \operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s dx$ when $b\,c-a\,d\neq 0$

2. $\int (g+h\,x)^m \operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right] dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{\operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{g+h\,x} dx$ when $b\,c-a\,d\neq 0$

Derivation: Integration by parts

Basis: $\frac{1}{g+h\,x} == \partial_x \frac{\operatorname{Log}[g+h\,x]}{h}$

Basis: $\partial_x \operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right] == \frac{b\,p\,r}{a+b\,x}+\frac{d\,q\,r}{c+d\,x}$

Rule: If $b\,c-a\,d\neq 0$, then

$$\int \frac{\operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{g+h\,x} dx \longrightarrow$$

$$\frac{\operatorname{Log}[g+h\,x]\operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{h}-\frac{b\,p\,r}{h}\int \frac{\operatorname{Log}[g+h\,x]}{a+b\,x} dx-\frac{d\,q\,r}{h}\int \frac{\operatorname{Log}[g+h\,x]}{c+d\,x} dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]/(g_.+h_.*x_),x_Symbol] :=
  Log[g+h*x]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]/h -
  b*p*r/h*Int[Log[g+h*x]/(a+b*x),x] -
  d*q*r/h*Int[Log[g+h*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r},x] && NeQ[b*c-a*d,0]
```

2: $\int (g+h x)^m \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right] dx$ when $b c-a d \neq 0 \land m \neq -1$

Derivation: Integration by parts

Basis: $(g+h x)^m = \partial_x \frac{(g+h x)^{m+1}}{h (m+1)}$

Basis: $\partial_x \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right] = \frac{b p r}{a+b x} + \frac{d q r}{c+d x}$

Rule: If $b c-a d \neq 0 \land m \neq -1$, then

$$\int (g+h x)^m \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right] dx \longrightarrow$$

$$\frac{(g+h x)^{m+1} \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]}{h (m+1)} - \frac{b p r}{h (m+1)} \int \frac{(g+h x)^{m+1}}{a+b x} dx - \frac{d q r}{h (m+1)} \int \frac{(g+h x)^{m+1}}{c+d x} dx$$

Program code:

```
Int[(g_.+h_.*x_)^m_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.],x_Symbol] :=
  (g+h*x)^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]/(h*(m+1)) -
  b*p*r/(h*(m+1))*Int[(g+h*x)^(m+1)/(a+b*x),x] -
  d*q*r/(h*(m+1))*Int[(g+h*x)^(m+1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,p,q,r},x] && NeQ[b*c-a*d,0] && NeQ[m,-1]
```

3. $\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^2}{g+h\,x}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^2}{g+h\,x}\,dx$ when $b\,c-a\,d\neq 0\,\land\,b\,g-a\,h==0$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left(\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]-\text{Log}\left[(a+b\,x)^{p\,r}\right]-\text{Log}\left[(c+d\,x)^{q\,r}\right]\right)==0$

Rule: If $b\,c-a\,d\neq 0\,\land\,b\,g-a\,h==0$, then

$$\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^2}{g+h\,x}\,dx\,\longrightarrow$$

$$\int \frac{\left(\text{Log}\left[(a+b\,x)^{p\,r}\right]+\text{Log}\left[(c+d\,x)^{q\,r}\right]\right)^2}{g+h\,x}\,dx+\left(\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]-\text{Log}\left[(a+b\,x)^{p\,r}\right]-\text{Log}\left[(c+d\,x)^{q\,r}\right]\right)\cdot$$
$$\left(2\int \frac{\text{Log}\left[(c+d\,x)^{q\,r}\right]}{g+h\,x}\,dx+\int \frac{\text{Log}\left[(a+b\,x)^{p\,r}\right]-\text{Log}\left[(c+d\,x)^{q\,r}\right]+\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{g+h\,x}\,dx\right)$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^2/(g_.+h_.*x_),x_Symbol] :=
  Int[(Log[(a+b*x)^(p*r)]+Log[(c+d*x)^(q*r)])^2/(g+h*x),x] +
  (Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]-Log[(a+b*x)^(p*r)]-Log[(c+d*x)^(q*r)])*
    (2*Int[Log[(c+d*x)^(q*r)]/(g+h*x),x] +
     Int[(Log[(a+b*x)^(p*r)]-Log[(c+d*x)^(q*r)]+Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r])/(g+h*x),x]) /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r},x] && NeQ[b*c-a*d,0] && EqQ[b*g-a*h,0]
```

x: $\int \frac{\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^2}{g+hx}\,dx$ when $bc-ad\neq 0\land bg-ah\neq 0\land dg-ch\neq 0$?? ??

Derivation: Piecewise constant extraction

Basis: $\partial_x\left(\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]-\text{Log}\left[(a+bx)^{pr}\right]-\text{Log}\left[(c+dx)^{qr}\right]\right)==0$

Rule: If $bc-ad\neq 0\land bg-ah==0$, then

$$\int \frac{\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^2}{g+hx}\,dx\longrightarrow$$

$$\int \frac{\left(\text{Log}\left[(a+bx)^{pr}\right]+\text{Log}\left[(c+dx)^{qr}\right]\right)^2}{g+hx}\,dx+$$
$$\left(\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]-\text{Log}\left[(a+bx)^{pr}\right]-\text{Log}\left[(c+dx)^{qr}\right]\right)$$
$$\int \frac{\text{Log}\left[(a+bx)^{pr}\right]+\text{Log}\left[(c+dx)^{qr}\right]+\text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]}{g+hx}\,dx$$

Program code:

```
(* Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^2/(g_.+h_.*x_),x_Symbol] :=
  Int[(Log[(a+b*x)^(p*r)]+Log[(c+d*x)^(q*r)])^2/(g+h*x),x] +
  (Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]-Log[(a+b*x)^(p*r)]-Log[(c+d*x)^(q*r)])*
    Int[(Log[(a+b*x)^(p*r)]+Log[(c+d*x)^(q*r)]+Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r])/(g+h*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r},x] && NeQ[b*c-a*d,0] && EqQ[b*g-a*h,0] *)
```

2: $\int \frac{\text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^2}{g+h x} \, dx$ when $b c - a d \neq 0$

Derivation: Integration by parts

Basis: $\frac{1}{g+h x} = \partial_x \frac{\text{Log}[g+h x]}{h}$

Basis: $\partial_x \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^2 = \frac{2 b p r \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]}{a+b x} + \frac{2 d q r \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]}{c+d x}$

Rule: If $b c - a d \neq 0$, then

$$\int \frac{\text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^2}{g+h x} \, dx \rightarrow$$

$$\frac{\text{Log}[g+h x] \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^2}{h} -$$

$$\frac{2 b p r}{h} \int \frac{\text{Log}[g+h x] \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]}{a+b x} \, dx - \frac{2 d q r}{h} \int \frac{\text{Log}[g+h x] \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]}{c+d x} \, dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^2/(g_.+h_.*x_),x_Symbol] :=
  Log[g+h*x]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^2/h -
  2*b*p*r/h*Int[Log[g+h*x]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]/(a+b*x),x] -
  2*d*q*r/h*Int[Log[g+h*x]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r},x] && NeQ[b*c-a*d,0]
```

4: $\int (g+h x)^m \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx$ when $b c-a d \neq 0 \land s \in \mathbb{Z}^+ \land m \neq -1$

Derivation: Integration by parts

Basis: $(g+h x)^m == \partial_x \frac{(g+h x)^{m+1}}{h (m+1)}$

Basis: $\partial_x \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s == \frac{b p r s}{a+b x} \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1} + \frac{d q r s}{c+d x} \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1}$

Rule: If $b c-a d \neq 0 \land s \in \mathbb{Z}^+ \land m \neq -1$, then

$$\int (g+h x)^m \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx \longrightarrow$$

$$\frac{(g+h x)^{m+1} \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{h (m+1)} -$$

$$\frac{b p r s}{h (m+1)} \int \frac{(g+h x)^{m+1} \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1}}{a+b x} dx - \frac{d q r s}{h (m+1)} \int \frac{(g+h x)^{m+1} \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1}}{c+d x} dx$$

Program code:

```
Int[(g_.+h_.*x_)^m_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_,x_Symbol] :=
  (g+h*x)^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(h*(m+1)) -
  b*p*r*s/(h*(m+1))*Int[(g+h*x)^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/(a+b*x),x] -
  d*q*r*s/(h*(m+1))*Int[(g+h*x)^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0] && NeQ[m,-1]
```

4. $\int \frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])^m\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^u}{j+k\,x}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])^m\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{j+k\,x}\,dx$ when $b\,c-a\,d\neq 0\,\land\,h\,j-g\,k==0\,\land\,m\in\mathbb{Z}^+$

Derivation: Integration by parts

Basis: If $h\,j-g\,k==0$, then $\frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])^m}{j+k\,x}==\partial_x\frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])^{m+1}}{k\,n\,t\,(m+1)}$

Basis: $\partial_x\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]==\frac{b\,p\,r}{a+b\,x}+\frac{d\,q\,r}{c+d\,x}$

Rule: If $b\,c-a\,d\neq 0\,\land\,h\,j-g\,k==0\,\land\,m\in\mathbb{Z}^+$, then

$$\int \frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])^m\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{j+k\,x}\,dx\,\longrightarrow$$

$$\frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])^{m+1}\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{k\,n\,t\,(m+1)}-$$

$$\frac{b\,p\,r}{k\,n\,t\,(m+1)}\int\frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])^{m+1}}{a+b\,x}\,dx-\frac{d\,q\,r}{k\,n\,t\,(m+1)}\int\frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])^{m+1}}{c+d\,x}\,dx$$

–

Program code:

```
Int[(s_.+t_.*Log[i_.*(g_.+h_.*x_)^n_.])^m_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]/(j_.+k_.*x_),x_Symbol] :=
  (s+t*Log[i*(g+h*x)^n])^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]/(k*n*t*(m+1)) -
  b*p*r/(k*n*t*(m+1))*Int[(s+t*Log[i*(g+h*x)^n])^(m+1)/(a+b*x),x] -
  d*q*r/(k*n*t*(m+1))*Int[(s+t*Log[i*(g+h*x)^n])^(m+1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,k,s,t,m,n,p,q,r},x] && NeQ[b*c-a*d,0] && EqQ[h*j-g*k,0] && IGtQ[m,0]
```

2: $\int \frac{\left(s+t\,\text{Log}\left[i\,(g+h\,x)^n\right]\right)\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{j+k\,x}\,dx$ when $b\,c-a\,d\neq 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left(\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]-\text{Log}\left[(a+b\,x)^{p\,r}\right]-\text{Log}\left[(c+d\,x)^{q\,r}\right]\right)=0$

Rule: If $b\,c-a\,d\neq 0$, then

$$\int \frac{\left(s+t\,\text{Log}\left[i\,(g+h\,x)^n\right]\right)\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{j+k\,x}\,dx\,\longrightarrow$$

$$\left(\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]-\text{Log}\left[(a+b\,x)^{p\,r}\right]-\text{Log}\left[(c+d\,x)^{q\,r}\right]\right)\int\frac{s+t\,\text{Log}\left[i\,(g+h\,x)^n\right]}{j+k\,x}\,dx+$$
$$\int\frac{\text{Log}\left[(a+b\,x)^{p\,r}\right]\,\left(s+t\,\text{Log}\left[i\,(g+h\,x)^n\right]\right)}{j+k\,x}\,dx+\int\frac{\text{Log}\left[(c+d\,x)^{q\,r}\right]\,\left(s+t\,\text{Log}\left[i\,(g+h\,x)^n\right]\right)}{j+k\,x}\,dx$$

Program code:

```
Int[(s_.+t_.*Log[i_.*(g_.+h_.*x_)^n_.])*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]/(j_.+k_.*x_),x_Symbol] :=
  (Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]-Log[(a+b*x)^(p*r)]-Log[(c+d*x)^(q*r)])*Int[(s+t*Log[i*(g+h*x)^n])/(j+k*x),x] +
  Int[(Log[(a+b*x)^(p*r)]*(s+t*Log[i*(g+h*x)^n]))/(j+k*x),x] +
  Int[(Log[(c+d*x)^(q*r)]*(s+t*Log[i*(g+h*x)^n]))/(j+k*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,k,s,t,n,p,q,r},x] && NeQ[b*c-a*d,0]
```

U: $\int \frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])^m\,\text{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]^u}{j+k\,x}\,dx$ when $b\,c-a\,d\neq 0$

Rule: If $b\,c-a\,d\neq 0$, then

$$\int \frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])^m\,\text{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]^u}{j+k\,x}\,dx \to \int \frac{(s+t\,\text{Log}[i\,(g+h\,x)^n])^m\,\text{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]^u}{j+k\,x}\,dx$$

Program code:

```
Int[(s_.+t_.*Log[i_.*(g_.+h_.*x_)^n_.])^m_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^u_./(j_.+k_.*x_),x_Symbol] :=
  Unintegrable[(s+t*Log[i*(g+h*x)^n])^m*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^u/(j+k*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,k,s,t,m,n,p,q,r,u},x] && NeQ[b*c-a*d,0]
```

6. $\int \frac{u\,\text{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]^s}{(a+b\,x)\,(c+d\,x)}\,dx$ when $b\,c-a\,d\neq 0\,\land\,p+q==0$

1: $\int \frac{\text{Log}\left[1+g\,\frac{a+b\,x}{c+d\,x}\right]\,\text{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]^s}{(a+b\,x)\,(c+d\,x)}\,dx$ when $b\,c-a\,d\neq 0\,\land\,s\in\mathbb{Z}^+\,\land\,p+q==0$

Derivation: Integration by parts

Basis: $\frac{\text{Log}\left[1+g\,\frac{a+b\,x}{c+d\,x}\right]}{(a+b\,x)\,(c+d\,x)} == -\partial_x\,\frac{\text{PolyLog}\left[2,-g\,\frac{a+b\,x}{c+d\,x}\right]}{b\,c-a\,d}$

Basis: If $p+q==0$, then $\partial_x\,\text{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]^s == \frac{p\,r\,s\,(b\,c-a\,d)}{(a+b\,x)\,(c+d\,x)}\,\text{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]^{s-1}$

Rule: If $b\,c-a\,d\neq 0\,\land\,s\in\mathbb{Z}^+\,\land\,p+q==0$, then

$$\int \frac{\text{Log}\left[1+g\,\frac{a+b\,x}{c+d\,x}\right]\,\text{Log}[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r]^s}{(a+b\,x)\,(c+d\,x)}\,dx \to$$

$$-\frac{\text{PolyLog}\left[2,\,-g\,\frac{a+b\,x}{c+d\,x}\right]\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{b\,c-a\,d}+p\,r\,s\int\frac{\text{PolyLog}\left[2,\,-g\,\frac{a+b\,x}{c+d\,x}\right]\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}}{(a+b\,x)\,(c+d\,x)}\,dx$$

–

Program code:

```
Int[u_*Log[v_]*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{g=Simplify[(v-1)*(c+d*x)/(a+b*x)],h=Simplify[u*(a+b*x)*(c+d*x)]},
  -h*PolyLog[2,1-v]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(b*c-a*d) +
  h*p*r*s*Int[PolyLog[2,1-v]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/((a+b*x)*(c+d*x)),x] /;
 FreeQ[{g,h},x]] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0] && EqQ[p+q,0]
```

2: $\int \frac{\text{Log}\left[i\left(j\,(g+h\,x)^t\right)^u\right]\,\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(c+d\,x)}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,p+q=0\,\wedge\,s\neq -1$

Derivation: Integration by parts

Basis: If $p+q=0$, then $\frac{\text{Log}\left[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r\right]^s}{(a+b\,x)\,(c+d\,x)} = \partial_x \frac{\text{Log}\left[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r\right]^{s+1}}{p\,r\,(s+1)\,(b\,c-a\,d)}$

Basis: $\partial_x \text{Log}\left[i\,\left(j\,(g+h\,x)^t\right)^u\right] = \frac{h\,t\,u}{g+h\,x}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,p+q=0\,\wedge\,s\neq -1$, then

$$\int \frac{\text{Log}\left[i\left(j\,(g+h\,x)^t\right)^u\right]\,\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(c+d\,x)}\,dx \longrightarrow$$

$$\frac{\text{Log}\left[i\left(j\,(g+h\,x)^t\right)^u\right]\,\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s+1}}{p\,r\,(s+1)\,(b\,c-a\,d)} - \frac{h\,t\,u}{p\,r\,(s+1)\,(b\,c-a\,d)}\int \frac{\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s+1}}{g+h\,x}\,dx$$

Program code:

```
Int[v_*Log[i_.*(j_.*(g_.+h_.*x_)^t_.)^u_.]*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{k=Simplify[v*(a+b*x)*(c+d*x)]},
  k*Log[i*(j*(g+h*x)^t)^u]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s+1)/(p*r*(s+1)*(b*c-a*d)) -
  k*h*t*u/(p*r*(s+1)*(b*c-a*d))*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s+1)/(g+h*x),x] /;
 FreeQ[k,x]] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,p,q,r,s,t,u},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && NeQ[s,-1]
```

3: $\int \frac{\text{PolyLog}\left[n,\,g\,\frac{a+b\,x}{c+d\,x}\right]\,\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(c+d\,x)}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,s\in\mathbb{Z}^+\,\wedge\,p+q=0$

Derivation: Integration by parts

Basis: $\frac{\text{PolyLog}\left[n,g\,\frac{a+b\,x}{c+d\,x}\right]}{(a+b\,x)\,(c+d\,x)} = \partial_x \frac{\text{PolyLog}\left[n+1,g\,\frac{a+b\,x}{c+d\,x}\right]}{b\,c-a\,d}$

Basis: If $p+q=0$, then $\partial_x \text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^s = \frac{p\,r\,s\,(b c-a d)}{(a+b x)\,(c+d x)} \text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^{s-1}$

Rule: If $b c-a d\neq 0 \land s\in\mathbb{Z}^+ \land p+q=0$, then

$$\int \frac{\text{PolyLog}\left[n, g\,\frac{a+b x}{c+d x}\right] \text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^s}{(a+b x)\,(c+d x)}\,dx \to$$

$$\frac{\text{PolyLog}\left[n+1, g\,\frac{a+b x}{c+d x}\right] \text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^s}{b c-a d} - p\,r\,s \int \frac{\text{PolyLog}\left[n+1, g\,\frac{a+b x}{c+d x}\right] \text{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^{s-1}}{(a+b x)\,(c+d x)}\,dx$$

Program code:

```
Int[u_*PolyLog[n_,v_]*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{g=Simplify[v*(c+d*x)/(a+b*x)],h=Simplify[u*(a+b*x)*(c+d*x)]},
  h*PolyLog[n+1,v]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(b*c-a*d) -
  h*p*r*s*Int[PolyLog[n+1,v]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/((a+b*x)*(c+d*x)),x] /;
 FreeQ[{g,h},x]] /;
FreeQ[{a,b,c,d,e,f,n,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0] && EqQ[p+q,0]
```

8: $\int \frac{\left(a+b\,\text{Log}\left[c\,\frac{\sqrt{d+e x}}{\sqrt{f+g x}}\right]\right)^n}{A+B x+C x^2}\,dx$ when $C d f-A e g=0 \land B e g-C\,(e f+d g)=0 \land n\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F[x] = 2\,(e f-d g)\,\text{Subst}\left[\frac{x}{\left(e-g x^2\right)^2} F\left[-\frac{d-f x^2}{e-g x^2}\right], x, \frac{\sqrt{d+e x}}{\sqrt{f+g x}}\right] \partial_x \frac{\sqrt{d+e x}}{\sqrt{f+g x}}$

Basis: If $C d f-A e g=0 \land B e g-C\,(e f+d g)=0$, then
$\frac{1}{A+B x+C x^2} = \frac{2 e g}{C\,(e f-d g)}\,\text{Subst}\left[\frac{1}{x}, x, \frac{\sqrt{d+e x}}{\sqrt{f+g x}}\right] \partial_x \frac{\sqrt{d+e x}}{\sqrt{f+g x}}$

Rule: If $C d f-A e g=0 \land B e g-C\,(e f+d g)=0 \land n\in\mathbb{Z}^+$, then

$$\int \frac{\left(a+b\,\text{Log}\left[c\,\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}\right]\right)^n}{A+B\,x+C\,x^2}\,dx \,\rightarrow\, \frac{2\,e\,g}{C\,(e\,f-d\,g)}\,\text{Subst}\left[\int \frac{(a+b\,\text{Log}[c\,x])^n}{x}\,dx,\,x,\,\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}\right]$$

–

Program code:

```
Int[(a_.+b_.*Log[c_.*Sqrt[d_.+e_.*x_]/Sqrt[f_.+g_.*x_]])^n_./(A_.+B_.*x_+C_.*x_^2),x_Symbol] :=
  2*e*g/(C*(e*f-d*g))*Subst[Int[(a+b*Log[c*x])^n/x,x],x,Sqrt[d+e*x]/Sqrt[f+g*x]] /;
FreeQ[{a,b,c,d,e,f,g,A,B,C,n},x] && EqQ[C*d*f-A*e*g,0] && EqQ[B*e*g-C*(e*f+d*g),0]
```

```
Int[(a_.+b_.*Log[c_.*Sqrt[d_.+e_.*x_]/Sqrt[f_.+g_.*x_]])^n_./(A_.+C_.*x_^2),x_Symbol] :=
  g/(C*f)*Subst[Int[(a+b*Log[c*x])^n/x,x],x,Sqrt[d+e*x]/Sqrt[f+g*x]] /;
FreeQ[{a,b,c,d,e,f,g,A,C,n},x] && EqQ[C*d*f-A*e*g,0] && EqQ[e*f+d*g,0]
```

9. $\int RF_x \operatorname{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]^s dx$

1: $\int RF_x \operatorname{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right] dx$ when $bc-ad\neq 0$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: $uA = uB + uC - (B+C-A)u$

Basis: $\partial_x\left(p\,r\operatorname{Log}[a+bx] + q\,r\operatorname{Log}[c+dx] - \operatorname{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]\right) = 0$

Rule: If $bc-ad\neq 0$, then

$$\int RF_x \operatorname{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right] dx \longrightarrow$$

$$p\,r\int RF_x \operatorname{Log}[a+bx]\,dx + q\,r\int RF_x \operatorname{Log}[c+dx]\,dx - \left(p\,r\operatorname{Log}[a+bx] + q\,r\operatorname{Log}[c+dx] - \operatorname{Log}\left[e\left(f(a+bx)^p(c+dx)^q\right)^r\right]\right)\int RF_x\,dx$$

Program code:

```
Int[RFx_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.],x_Symbol] :=
  p*r*Int[RFx*Log[a+b*x],x] +
  q*r*Int[RFx*Log[c+d*x],x] -
  (p*r*Log[a+b*x]+q*r*Log[c+d*x] - Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r])*Int[RFx,x] /;
FreeQ[{a,b,c,d,e,f,p,q,r},x] && RationalFunctionQ[RFx,x] && NeQ[b*c-a*d,0] &&
  Not[MatchQ[RFx,u_.*(a+b*x)^m_.*(c+d*x)^n_. /; IntegersQ[m,n]]]
```

x: $\int RF_x \,\text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right] dx$ when $b c - a d \neq 0$

Derivation: Integration by parts

Basis: $\partial_x \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right] == \frac{b p r}{a+b x} + \frac{d q r}{c+d x}$

Rule: If $b c - a d \neq 0$, let $u \to \int RF_x \, dx$, then

$$\int RF_x \,\text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right] dx \longrightarrow u \,\text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right] - b p r \int \frac{u}{a+b x} dx - d q r \int \frac{u}{c+d x} dx$$

Program code:

```
(* Int[RFx_*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.],x_Symbol] :=
  With[{u=IntHide[RFx,x]},
  u*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r] - b*p*r*Int[u/(a+b*x),x] - d*q*r*Int[u/(c+d*x),x] /;
 NonsumQ[u]] /;
FreeQ[{a,b,c,d,e,f,p,q,r},x] && RationalFunctionQ[RFx,x] && NeQ[b*c-a*d,0] *)
```

2: $\int RF_x \operatorname{Log}\left[e\left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx$ when $s \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $s \in \mathbb{Z}^+$, then

$$\int RF_x \operatorname{Log}\left[e\left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx \longrightarrow \int \operatorname{Log}\left[e\left(f (a+b x)^p (c+d x)^q\right)^r\right]^s \text{ExpandIntegrand}[RF_x, x]\, dx$$

Program code:

```
Int[RFx_*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{u=ExpandIntegrand[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s,RFx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && RationalFunctionQ[RFx,x] && IGtQ[s,0]
```

U: $\int RF_x \operatorname{Log}\left[e\left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx$

Rule:

$$\int RF_x \operatorname{Log}\left[e\left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx \longrightarrow \int RF_x \operatorname{Log}\left[e\left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx$$

Program code:

```
Int[RFx_*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  Unintegrable[RFx*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s,x] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && RationalFunctionQ[RFx,x]
```

N: $\int u \operatorname{Log}\left[e \left(f v^p w^q\right)^r\right]^s dx$ when $v == a + b x \land w == c + d x$

Derivation: Algebraic normalization

– Rule: If $v == a + b x \land w == c + d x$, then

$$\int u \operatorname{Log}\left[e \left(f v^p w^q\right)^r\right]^s dx \longrightarrow \int u \operatorname{Log}\left[e \left(f (a + b x)^p (c + d x)^q\right)^r\right]^s dx$$

– Program code:

```
Int[u_.*Log[e_.*(f_.*v_^p_.*w_^q_.)^r_.]^s_.,x_Symbol] :=
  Int[u*Log[e*(f*ExpandToSum[v,x]^p*ExpandToSum[w,x]^q)^r]^s,x] /;
FreeQ[{e,f,p,q,r,s},x] && LinearQ[{v,w},x] && Not[LinearMatchQ[{v,w},x]] && AlgebraicFunctionQ[u,x]
```

```
Int[u_.*Log[e_.*(f_.*(g_+v_./w_))^r_.]^s_.,x_Symbol] :=
  Int[u*Log[e*(f*ExpandToSum[v+g*w,x]/ExpandToSum[w,x])^r]^s,x] /;
FreeQ[{e,f,g,r,s},x] && LinearQ[w,x] && (FreeQ[v,x] || LinearQ[v,x]) && AlgebraicFunctionQ[u,x]
```

x: $\int \frac{\operatorname{Log}\left[i \left(j (g + h x)^s\right)^t\right] \operatorname{Log}\left[e \left(f (a + b x)^p (c + d x)^q\right)^r\right]}{m + n x} dx$

Derivation: Integration by substitution

Basis: $F[x] == \frac{1}{n} \operatorname{Subst}\left[F\left[\frac{x-m}{n}\right], x, m + n x\right] \partial_x (m + n x)$

– Rule:

$$\int \frac{\operatorname{Log}\left[i \left(j (g + h x)^s\right)^t\right] \operatorname{Log}\left[e \left(f (a + b x)^p (c + d x)^q\right)^r\right]}{m + n x} dx \longrightarrow$$

$$\frac{1}{n}\,\text{Subst}\left[\int \frac{\text{Log}\left[i\left(j\left(-\frac{h\,m-g\,n}{n}+\frac{h\,x}{n}\right)^{s}\right)^{t}\right]\,\text{Log}\left[e\left(f\left(-\frac{b\,m-a\,n}{n}+\frac{b\,x}{n}\right)^{p}\left(-\frac{d\,m-c\,n}{n}+\frac{d\,x}{n}\right)^{q}\right)^{r}\right]}{x}\,dx,\ x,\ m+n\,x\right]$$

Program code:

```
(* Int[Log[g_.*(h_.*(a_.+b_.*x_)^p_.)^q_.]*Log[i_.*(j_.*(c_.+d_.*x_)^r_.)^s_.]/(e_+f_.*x_),x_Symbol] :=
  1/f*Subst[Int[Log[g*(h*Simp[-(b*e-a*f)/f+b*x/f,x]^p)^q]*Log[i*(j*Simp[-(d*e-c*f)/f+d*x/f,x]^r)^s]/x,x],x,e+f*x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,p,q,r,s},x] *)
```